pmaxima

```
Maxima 5.30.0 http://maxima.sourceforge.net
using Lisp GNU Common Lisp (GCL) GCL 2.6.7 (a.k.a. GCL)
Distributed under the GNU Public License. See the file COPYING.
Dedicated to the memory of William Schelter.
```

Exercice

Le plan est muni d'un repère orthonormé.
On appelle (P) la parabole dont une représentation paramétrique est :

$$x = \frac{t^2}{2}, \quad , y = t, \quad t \in \mathbf{R}.$$

Dans la suite on note M le point de (P) d'ordonnée t et A le point de (P) d'ordonnée -1.

1/ Déterminer une équation de la tangente (T) en M à la parabole (P).

2/ Déterminer une équation de la droite (D) passant par A et perpendiculaire à (T).

3/ Démontrer que le point d'intersection H de ces deux droites a pour coordonnées :

$$x = -\frac{t}{1+t^2}, \quad y = \frac{t^3+t-2}{2(1+t^2)}.$$

4/ On appelle (Γ) la courbe paramétrée décrite par H.

a) Calculer les coordonnées du vecteur $\frac{\mathrm{d}\overrightarrow{H}}{\mathrm{d}t}$.

b) On suppose que M est distinct de A. Démontrer que le cercle de diamètre $[AM]$ et la courbe (Γ) sont tangentes en H.

Définition de la fonction fixant les coordonnées des points de (P).

```
▷ f(t):=[t^2/2,t]$
```

Définition de (P).

```
▷ P:ev(Courbe(f(t),t,-3,3),NUMPOINTS:25)$
```

Définition de A.

```
▷ A:Pair2Point(f(-1));
```

$$\mathit{point}\left(\frac{1}{2}, -1\right)$$

Choix d'un point M de (P) (la valeur de t choisie pour la construction est représentée par `tt`).

```
▷ tt:2.7$
```

```
▷ M:Pair2Point(f(tt));
```

$$\mathit{point}(3.645, 2.7)$$

Première figure

```
▷ Figure('P, 'A, 'M);
```



Détermination de l'équation de la tangente (T) en M.

```
▷ vd(t):=diff(f(t),t);
```

```
▷ eqT:determinant(matrix([x,y]-f(t),vd(t)))=0;
```

$$-t\ (y-t)+x-\frac{t^2}{2}=0$$

Détermination de l'équation de la droite (D) passant par A et perpendiculaire à (T).

```
▷ eqD:([x,y]-f(-1)).vd(t)=0;
```

$$y+t\ \left(x-\frac{1}{2}\right)+1=0$$

Intersection de D et T : le point H.

```
▷ Ht:solve([eqT,eqD],[x,y])[1];
```

$$\left[x=-\frac{t}{t^2+1}, y=\frac{t^3+t-2}{2\,t^2+2}\right]$$

Nous retrouvons les expressions de l'énoncé. Exprimons ce résultat en terme de coordonnées :

```
▷ Ht:subst(Ht,[x,y]);
```

$$\left[-\frac{t}{t^2+1}, \frac{t^3+t-2}{2\,t^2+2}\right]$$

Dans le même temps définissons la fonction exprimant les coordonnées du point courant de (Γ) (lieu des points H).

```
▷ define(g(t),Ht);
```

Deuxième figure

```
▷ D:DroiteEq(subst(t=tt,eqD))$
```

▷ `T:DroiteEq(subst(t=tt,eqT))$`

▷ `H:Pair2Point(g(tt))$`

▷ `Figure('D,'T,'H);`



Troisième figure

Introduction de la courbe (Γ).

▷ `G:ev(Courbe(g(t),t,-9,9),NUMPOINTS:100)$`

▷ `Figure('G);`



Détermination de $\frac{\mathrm{d}\overrightarrow{H}}{\mathrm{d}t}$.

```
▷ (vdh(t):=diff(g(t),t), ratsimp(vdh(t)));
```

$$\left[\frac{t^2-1}{t^4+2t^2+1}, \frac{t^4+2t^2+4t+1}{2t^4+4t^2+2}\right]$$

Détermination de la tangente à (Γ) en H.

```
▷ eqG:num(ratsimp(determinant(matrix([x,y]-g(t),vdh(t)))))=0;
```

$$-\left(2t^2-2\right)y-\left(-t^4-2t^2-4t-1\right)x+2t^3+2=0$$

Le cercle de diamètre $[AM]$ passe par H, par construction de H. Nous pouvons vérifier que ce cercle est tangent, en H, à (Γ) [1] en effectuant le produit scalaire entre $\frac{\mathrm{d}\overrightarrow{H}}{\mathrm{d}t}$ et $\overrightarrow{IM}$ où I est le milieu de $[AM]$.

```
▷ ratsimp(vdh(t).(g(t)-(f(-1)+f(t))/2));
```

$$0$$

Quatrième figure

```
▷ C:CercleDiametre(A,M)$
```

```
▷ T2:DroiteEq(subst(t=tt,eqG))$
```

```
▷ S:Segment(A,M)$
```

```
▷ Figure('C,'T2,'S);
```



1. En fait tangent à la tangente à (Γ) en H.